Vectors in the Plane

```
> restart:with(plots):
```

We need to input the VectorCalculus package to get the vector commands we need.

```
> with(VectorCalculus);
```

[&x, `*`, `+`, `-`, `.`, <,>, <|>, About, AddCoordinates, ArcLength, BasisFormat, Binormal, Compatibility, ConvertVector, CrossProd, CrossProduct, Curl, Curvature, D, Del, DirectionalDiff, Divergence, DotProd, DotProduct, Flux, GetCoordinateParameters, GetCoordinates, GetNames, GetPVDescription, GetRootPoint, GetSpace, Gradient, Hessian, IsPositionVector, IsRootedVector, IsVectorField, Jacobian, Laplacian, LineInt, MapToBasis, Nabla, Norm, Normalize, PathInt, PlotPositionVector, PlotVector, PositionVector, PrincipalNormal, RadiusOfCurvature, RootedVector, ScalarPotential, SetCoordinateParameters, SetCoordinates, SpaceCurve, SurfaceInt, TNBFrame, Tangent, TangentLine, TangentPlane, TangentVector, Torsion, Vector, VectorField, VectorPotential, VectorSpace, Wronskian, diff, eval, evalVF, int, limit, series]

Defining free vectors (arrows), rooted vectors, and position vectors.

Consider a **free vector** to be the equivalence class of all **arrows** having the same direction and magnitude, i.e. an arrow and all of its parallel translates. So that we can enter vectors using standard vector notation, we set BasisFormat to **false**.

```
> BasisFormat(false);
```

true

We can enter free vectors as either **column** (the default) or **row** vectors. There are two ways to enter **column** vectors.

```
> v1:=<3,5>;
```

$$v1 := \begin{bmatrix} 3 \\ 5 \end{bmatrix}$$

```
> v2:=Vector([3,5]);
```

$$v2 := \begin{bmatrix} 3 \\ 5 \end{bmatrix}$$

We enter row vectors as follows.

```
> v3:=<3|5>;
```

$$v3 := \begin{bmatrix} 3 & 5 \end{bmatrix}$$

We use the About statement to gain information about our vectors.

```
> About(v1);About(v2);About(v3);
```

$$\begin{bmatrix} \textit{Type:} & \textit{Free Vector} \\ \textit{Components:} & [3, 5] \\ \textit{Coordinates:} & \textit{cartesian} \end{bmatrix}$$

$$\begin{bmatrix} \textit{Type:} & \textit{Free Vector} \\ \textit{Components:} & [3, 5] \\ \textit{Coordinates:} & \textit{cartesian} \end{bmatrix}$$

$$\begin{bmatrix} \textit{Type:} & \textit{Free Vector} \\ \textit{Components:} & [3, 5] \\ \textit{Coordinates:} & \textit{cartesian} \end{bmatrix}$$

Free vectors are plotted using PlotVector the command.

```
> PlotVector(v1);
```



We can adjust the shape of the arrow by using parameters.

```
> PlotVector(v1,width=.1,head_width=.2,head_length=.4,color=blue);
```



If **head_length** is not given, the head is about $\frac{1}{5}$ the length of the arrow.

```
> PlotVector(v1,width=.1,head_width=.2,color=cyan);
```



A **rooted vector** is an arrow whose tail is fixed to a point in the plane. For instance, place the tail of **v1** at the point (-2,-3).

```
> PlotVector(<-2,-3>,v1,width=.1,head_width=.2,head_length=.4,
  color=green);
```



We define and plot a **rooted vector** using RootedVector as follows.

```
> v4:=RootedVector(root=[-2,-3],[3,5]);
```

$$v4 := \begin{bmatrix} 3 \\ 5 \end{bmatrix}$$

```
> About(v4);
```

$$\begin{bmatrix} \textit{Type:} & \textit{Rooted Vector} \\ \textit{Components:} & [3, 5] \\ \textit{Coordinates:} & \textit{cartesian} \\ \textit{Root Point:} & [-2, -3] \end{bmatrix}$$

```
> PlotVector(v4,width=.1,head_width=.2,head_length=.4,color=orange)
;
```



We define and plot a position vector, a vector with tail at the origin, using PositionVector as follows.

```
> v5:=PositionVector([5,1]);
```

$$v5 := \begin{bmatrix} 5 \\ 1 \end{bmatrix}$$

```
> About(v5);
```

$$\begin{bmatrix} \mathit{Type:} & \mathit{Position\ Vector} \\ \mathit{Components:} & [5, 1] \\ \mathit{Coordinates:} & \mathit{cartesian} \\ \mathit{Root\ Point:} & [0, 0] \end{bmatrix}$$

```
> PlotVector(v5,width=.1,head_width=.2,head_length=.4,color=violet)
;
```



We get the extra fat arrow since the axes have diffent scales. We can set **scaling** to **constrained** as a plot option to get equally scaled axes.

```
> PlotVector(v5,width=.1,head_width=.2,head_length=.4,color=violet,
  scaling=constrained);
```



Basic Operations for Vectors.

```
> restart:with(plots):with(VectorCalculus):BasisFormat(false);
```

true

We first define six (column) vectors: two **free**, two **rooted**, and two **position**.

```
> v1:=<3,5>;v2:=<-6,2>;v3:=RootedVector(root=[-2,1],[-4,2]);v4:=
  RootedVector(root=[-2,1],[5,3]);v5:=PositionVector([5,1]);v6:=
  PositionVector([-3,2]);
```

$$v1 := \begin{bmatrix} 3 \\ 5 \end{bmatrix}$$

$$v2 := \begin{bmatrix} -6 \\ 2 \end{bmatrix}$$

$$v3 := \begin{bmatrix} -4 \\ 2 \end{bmatrix}$$

$$v4 := \begin{bmatrix} 5 \\ 3 \end{bmatrix}$$

$$v5 := \begin{bmatrix} 5 \\ 1 \end{bmatrix}$$

$$v6 := \begin{bmatrix} -3 \\ 2 \end{bmatrix}$$

We use the Norm command to find the **norm** or **magnitude** of the three vectors.

```
> Norm(v1);Norm(v2);Norm(v3);Norm(v4);Norm(v5);Norm(v6);
```

$$\sqrt{34}$$

$$2\sqrt{10}$$

$$2\sqrt{5}$$

$$\sqrt{34}$$

$$\sqrt{26}$$

$$\sqrt{13}$$

Vector addition is done by component. We first add two free vectors and then two position vectors.

```
> v12:=v1+v2;About(v12);v56:=v5+v6;About(v56);
```

$$v12 := \begin{bmatrix} -3 \\ 7 \end{bmatrix}$$

$$\begin{bmatrix} \mathit{Type:} & \mathit{Free\ Vector} \\ \mathit{Components:} & [-3, 7] \\ \mathit{Coordinates:} & \mathit{cartesian} \end{bmatrix}$$

$$v56 := \begin{bmatrix} 2 \\ 3 \end{bmatrix}$$

$$\begin{bmatrix} \mathit{Type:} & \mathit{Position\ Vector} \\ \mathit{Components:} & [2, 3] \\ \mathit{Coordinates:} & \mathit{cartesian} \\ \mathit{Root\ Point:} & [0, 0] \end{bmatrix}$$

We see the sums are of the same type as the addends. Rooted vectors can only be added if their roots (points of origin) are the same with the sum having the same root..

```
> v34:=v3+v4;About(v34);
```

$$v34 := \begin{bmatrix} 1 \\ 5 \end{bmatrix}$$

$$\begin{bmatrix} \textit{Type:} & \textit{Rooted Vector} \\ \textit{Components:} & [1, 5] \\ \textit{Coordinates:} & \textit{cartesian} \\ \textit{Root Point:} & [-2, 1] \end{bmatrix}$$

What about adding a free vector or a position vector to a rooted vector?

```
> v13:=v1+v3;About(v13);
```

$$v13 := \begin{bmatrix} -1 \\ 7 \end{bmatrix}$$

$$\begin{bmatrix} \textit{Type:} & \textit{Rooted Vector} \\ \textit{Components:} & [-1, 7] \\ \textit{Coordinates:} & \textit{cartesian} \\ \textit{Root Point:} & [-2, 1] \end{bmatrix}$$

```
> v35:=v3+v5;About(v35);
```

$$v35 := \begin{bmatrix} 1 \\ 3 \end{bmatrix}$$

$$\begin{bmatrix} \textit{Type:} & \textit{Rooted Vector} \\ \textit{Components:} & [1, 3] \\ \textit{Coordinates:} & \textit{cartesian} \\ \textit{Root Point:} & [-2, 1] \end{bmatrix}$$

We see the sum is a rooted vector with the same root as the rooted addend. What about a free vector with a position vector?

```
> v15:=v1+v5;About(v15);
```

$$v15 := \begin{bmatrix} 8 \\ 6 \end{bmatrix}$$

$$\begin{bmatrix} \textit{Type:} & \textit{Free Vector} \\ \textit{Components:} & [8, 6] \\ \textit{Coordinates:} & \textit{cartesian} \end{bmatrix}$$

The sum is a free vector. We next look at **scalar multiplication**. We first multiply by $\frac{1}{2}$.

```
> w1:=(1/2)*v1;w3:=(1/2)*v3;w5:=(1/2)*v5;
```

$$w1 := \begin{bmatrix} \frac{3}{2} \\ \frac{5}{2} \end{bmatrix}$$

$$w3 := \begin{bmatrix} -2 \\ 1 \end{bmatrix}$$

$$w5 := \begin{bmatrix} \frac{5}{2} \\ \frac{1}{2} \end{bmatrix}$$

We view the original vectors in **red** and the scalar multiples in **blue**.

```
> p4:=PlotVector([v1,v3,v5],width=.1,head_width=.2,head_length=.4,
  scaling=constrained,color=red):
  p5:=PlotVector([w1,w3,w5],width=.2,head_width=.4,head_length=.4,
  scaling=constrained,color=blue):
  p6:=textplot({[3,4,"v1"],[-5,2,"v3"],[1.2,1.1,"w1"],[-3,1,"w3"],
  [4,.5,"v5"],[1.6,-.2,"w5"]},font=[TIMES,BOLD,14]):
  display(p4,p5,p6);
```



Next we multiply by -2.

```
> z1:=(-2)*v1;z3:=(-2)*v3;z5:=(-2)*v5;
```

$$z1 := \begin{bmatrix} -6 \\ -10 \end{bmatrix}$$

$$z3 := \begin{bmatrix} 8 \\ -4 \end{bmatrix}$$

$$z5 := \begin{bmatrix} -10 \\ -2 \end{bmatrix}$$

Again we visualize the original vectors in **red** and the scalar multiples in **blue**.

```
> p7:=PlotVector([v1,v3,v5],width=.1,head_width=.2,head_length=.4,
  scaling=constrained,color=red):
  p8:=PlotVector([z1,z3,z5],width=.1,head_width=.2,head_length=.4,
  scaling=constrained,color=blue):
  p9:=textplot({[3,4,"v1"],[-5,2,"v3"],[-6,-8,"z1"],[5,-2,"z3"],[4,
  .5,"v5"],[-8,-1,"z5"]},font=[TIMES,BOLD,14]):
  display(p7,p8,p9);
```



Horizontal and vertical components of vectors.

```
> restart:with(plots):with(plottools):with(VectorCalculus)
  :BasisFormat(false);
```

true

We enter the standard vectors.

```
> i:=PositionVector([1,0]);j:=PositionVector([0,1]);
```

$$i := \begin{bmatrix} 1 \\ 0 \end{bmatrix}$$

$$j := \begin{bmatrix} 0 \\ 1 \end{bmatrix}$$

Now we view the decomposition <3, 2> = 3**i** + 2**j** in two dimensions.

```
> p7:=PlotVector([i,j],width=.1,head_width=.2,head_length=.2,
  scaling=constrained,color=blue):
  p8:=PlotVector([3*i,RootedVector(root=3*i,[0,2]),3*i+2*j],width=
  .2,head_width=.4,head_length=.4,scaling=constrained,color=red):
  p9:=textplot({[.5,-.2,"i"],[-.4,.5,"j"],[2,-.4,"3*i"],[3.4,.9,"2*
  j"],[1,1.4,"2*i+3*j"]},font=[TIMES,BOLD,14]):
  display(p7,p8,p9);
```